
ABSTRACT 
Data quality is essential to proper planning, evaluation, and research in nearly all industries. This paper describes 
how PROC FORMAT and PROC REPORT can be employed to generate a color-coded report that visibly displays 
sources of data values outside acceptable ranges. 
 
Using birth certificate data collected and submitted by hospitals, this report examines multiple data fields against 
specified criteria and accordingly assigns different colors. By glancing at the report, the reader can quickly identify 
hospitals with reporting problems in several areas or if there is a particular item that is a problem for many hospitals. 
 
INTRODUCTION 
The excellence and completeness of birth certificate data is vital in public health research, planning and services. 
State and Federal Maternal and Child Health (MCH) programs utilize birth certificate data to target mothers from 
particular racial/ethnic and age groups with low utilization of prenatal care or to investigate maternal characteristics 
associated with low birth weight, prematurity, infant mortality and other unfavorable birth outcomes.   
 
Arkansas State laws require birth certificates to be completed for all births, and Federal law mandates national 
collection and publication of births and other vital statistics data1. The National Vital Statistics System (NVSS), the 
Federal compilation of this data, is the result of cooperation between CDC’s National Center for Health Statistics 
(NCHS) and state agencies that brings about a wealth of statistical information from birth certificates. 
 
NCHS, every year, sets up standards of acceptable proportions of poor data fields reported as not classifiable (i.e., 
outside predetermined ranges), unknown or blank in birth certificate data. Since the vast majority of birth certificates 
are collected and submitted by hospitals, the Center for Health Statistics at the Arkansas Division of Health needed a 
report that can easily identify fields with high proportions of invalid data by hospital of birth. Such a report is important 
for administrators of vital records and field representatives who can contact these hospitals to provide educational 
seminars to nurses and clerks on appropriate methods in collecting and reporting birth data. 
 
With the introduction of ODS, we wanted to take advantage of its strong reporting capabilities and the ability to create 
appealing tables, to craft a report that can be used to identify hospitals that contribute to high quantities of poor data. 
 
THE ASSIGNMENT   
Using birth data submitted by 54 states and territories, NCHS produces an annual state-level report entitled: 
“NATALITY NOT CLASSIFIABLE AND UNKNOWN” and makes it available to Centers for Health Statistics and Vital 
Records, or any other individual state agencies that collect, edit, and publish vital statistics data.  
 
In 2003, NCHS selected 40 data fields to report on. The report includes the acceptable percentage of records 
(standard) that are not classifiable, blank and unknown. Standards are computed using criteria of multiplying the 
nationwide median for 1998 for a particular item, as a percentage of records received by NCHS, by a multiplier. The 
multiplier was 1.5 for 2003, which is 0.5 less than that of 2002. The following is an example of how the standard was 
computed for prenatal visits in different years. 
 

Prenatal 
Visits 

1998 Median (%) 2000 Standard = 
3X the Median 
for 1998 

2001 Standard = 
2.5X the Median 
for 1998 

2002 Standard = 
2X the Median 
for 1998 

2003 Standard = 
1.5X the Median 
for 1998 

 1.99 5.97 4.98 3.98 2.99 
 
Since most births occur in hospitals and freestanding birth facilities, which are required to submit birth certificate data 
to the state health department, we wanted to generate an Arkansas-specific report by hospital of birth using the same 
standards supplied by NCHS. We also wanted to identify fields in which Arkansas exceeds the NCHS standard for 
that particular year, as well as to distinguish hospitals that exceed that standard. One senior analyst suggested that 
the report also highlights fields that may potentially exceed the standard for both the state and individual hospitals.   

For purposes of this paper, we have selected 8 birth certificate items that are looked at more often in analyzing 
preterm birth∗. These are: clinical estimate of gestation (gest), day last normal menses began (menda), month last 
menses began (menmo), year last menses began (menyr), month prenatal care began (care), named month prenatal 
care began (nammo), prenatal visits (visits), and weight gain during pregnancy (wtgain).  
 
BEFORE ODS 
Before we learned about ODS, we generated a similar report but used Microsoft Excel to present the tables.  We 
categorized the selected variables, summarized the data using PROC SUMMARY and exported the data table to 
Excel. In Excel, we customized some Visual Basic for Applications (VBA) scripts generated by running user-defined 
Excel macros to color-shade cells where values exceeded the standards. With the introduction of ODS and the 
powerful PROC REPORT, we thought of generating the report “by clicking only one button” entirely in a SAS 
environment.  
 
THE SAS PROGRAM         
 
GETTING STARTED 
 
   OPTIONS PAGENO = 1 CENTER PAPERSIZE = LEGAL ORIENTATION = LANDSCAPE;   
   LIBNAME birth 'path-to-server/birth database'; 
   %LET data = birth.birth2003;  
   %LET path = C:\SUGI31;   
 
To be able to generate a report with all your summarized variables (numbers and percents) lined up in one page, use 
PAPERSIZE = LEGAL and ORINETATION = LANDSCAPE options in the OPTIONS statement . The %LET macro-
program statement , the simplest way to create and assign a value to a macro variable, is used to assign a value to 
the permanent birth data file and %LET statement  defines the path where you want to store your generated report 
(in a PDF file format, as discussed later). 
 
   DATA temp (KEEP = ostate bhosp_ bloc gest gest1 menda menda1 menmo 
                     menmo1 menyr menyr1 care care1 nammo nammo1  
                     visits visits1 wtgain_ wtgain1 rec_err);  
      SET &data (PW = *****); 
      IF bloc NOT IN ('1', '2') THEN bhosp_ = 'Non-Hosp';         
 
      IF NOT ('02' LE gest LE '44') THEN gest1 = 1;  
      ELSE gest1 = 0;  
      IF NOT ('01' LE menda LE '31') THEN menda1 = 1;  
      ELSE menda1 = 0; 
      IF NOT ('01' LE menmo LE '12') THEN menmo1 = 1;  
      ELSE menmo1 = 0;  
      IF menyr IN ('9999', ' ', '0000') THEN menyr1 = 1;  
      ELSE menyr1 = 0; 
      IF NOT ('00' LE care LE '09') THEN care1 = 1;                           
      ELSE care1 = 0;                                                  
      /*CODE ONLY IF "MONTH PRENATAL CARE BEGAN" IS BLANK*/                
      IF care EQ '' and NOT ('01' LE nammo LE '12') THEN nammo1 = 1;    
      ELSE nammo1 = 0; 
      IF NOT ('00' LE visits LE '97') THEN visits1 = 1;  
      ELSE visits1 = 0; 
      IF wtgain_ IN (999,.) OR wtgain_ GT 200 THEN wtgain1 = 1;  
      ELSE wtgain1 = 0; 
      IF gest1 + menda1 + menmo1 + menyr1 +  
         care1 + nammo1 + visits1 + wtgain1 gt 0 THEN rec_err = 1; 
      ELSE rec_err = 0; 
   RUN; 
 
DATA step  creates a temporary analysis data set and the KEEP =, a data step option, reduces the number of 
variables to these that you will be using throughout the program in the work dataset. The IF-THEN statement  
recodes birth hospital numbers for non-hospital births to ‘Non-Hosp’. Evaluating this type of birth certificates uncovers 
                                                 
∗ Arkansas has been experiencing a decline in preterm birth rate in the last five years. The accuracy and completeness of pertinent 
data is essential to analyzing the significance of the trend. 

the quality of mostly home-birth certificate data. IF-THEN/ELSE group of arguments  creates new analysis variables 
from original birth variables of interest. In these newly created variables, valid data entries (within an acceptable 
range) will be assigned 0s, while not-classifiables, unknowns, blanks, and other invalid entries will be assigned 1s. 
Note that the last IF-THEN/ELSE argument creates a new variable (rec_err), which is coded 1 if a birth record 
contains at least one invalid entry in the selected fields. Clean birth records are coded 0.  
 
STEP 1: SUMMARIZE YOUR DATA USING THE SUMMARY PROCEDURE 
 
   PROC SUMMARY DATA = temp;  
      WHERE ostate EQ '004';    
      CLASS bhosp_;   
      VAR gest1 -- rec_err;   
      OUTPUT OUT = summary (DROP = _TYPE_ RENAME = (_FREQ_ = Birth))  
      SUM = MEAN = / AUTONAME;  
   RUN; 
 
As the name implies, the “good old” SUMMARY procedure can be used to summarize data by computing counts and 
percentages. PROC SUMMARY statement  invokes the procedure and the WHERE statement  requests that the 
summarization be limited to births that occurred in Arkansas, as we are interested in instate birth hospitals only. Since 
we need to summarize the data at the hospital level, variable bhosp_ (birth hospital code) is inserted as a class-
variable in the CLASS statement. The range of variables listed in the VAR statement  represents the 8 selected 
birth certificate items for the report together with the newly created rec_err (true if record contains at least one invalid 
entry) - a total of 9 variables. 
 
Request the output of the SUMMARY procedure be routed to a work data set (summary) by using the OUTPUT OUT 
= statement .  When you output your summarized results into a SAS data set, PROC SUMMARY, by default, 
generates two variables: _TYPE_, and _FREQ_. Values in variable _TYPE_ correspond to the different levels of 
summarization requested in PROC SUMMARY, which depend on the number of variables in the CLASS statement. If 
you have one variable in the CLASS statement, variable _TYPE_ will have 2 values: 0 and 1s, in which the 0 
represents the grand total  (count, average, or other requested statistics) and the 1s denote the totals for each value 
in the CLASS-variable. Similarly, if you have two variables in the CLASS statement, the 4 values of variable _TYPE_ 
will range from 0, 1s, 2s and 3s. In this scenario, 0 would represent the overall total; the 1s would signify the totals for 
each value in the first variable in the CLASS statement; the 2s would stand for the totals for each value in the second 
variable in the CLASS statement; and finally, the 3s would represent the crosstabulation of values in the first and 
second variables in the CLASS statement. 
 
The automatically generated variable _TYPE_ is most beneficial when you specify two or more variables in the 
CLASS statement. Understanding each level of summarization and the value of _TYPE_ that corresponds to that 
level enables you to control what level of analysis you want to show in your report (i.e., state, region, county). In this 
example, we are planning to present the state overall level (_TYPE_  = 0), as well as the hospital level (_TYPE_  = 
1). Since there is no need to keep variable  _TYPE_, use the DROP = option to suppress it from the outputted 
dataset summary.  
 
The second automatically generated variable in PROC SUMMARY (_FREQ_) indicates the frequency or count of 
records for each level in the CLASS variable. In our example, it indicates total births at the state (_TYPE_ = 0) and 
hospital (_TYPE_ = 1) levels. You can use the RENAME = option to rename this variable to Birth1. The SUM = and 
MEAN = options are the requested statistics for the variables listed in the VAR statement . They compute the 
counts and proportions, respectively. Note that the word “mean” may indicate statistical average but will work in this 
scenario since the data is categorical (1s or 0s). The AUTONAME option assigns the text strings “_SUM” and 
“_MEAN” to generated variable names containing computed frequencies and proportions. We could have used the 
AUTOLABEL option as well, but the flexibility of DEFINE statements in PROC REPORT (discussed later in this 
paper) gives you more control on the appearance of column headers.     
 
STEP 2: ROUND AND SORT YOUR RESULTS IN A DATA STEP 
 
   DATA report1 (DROP = i); 
      SET summary; 
      ARRAY new (9) gest1_mean -- rec_err_mean;   
      DO i = 1 TO 9;  
      new (i) = ROUND (new (i), 0.0001)*100;  

                                                 
1 Or, you can leave the variable _FREQ_ in dataset “summary” in step 1 and label it using DEFINE statement in step 5 

      END; 
      IF bhosp_ EQ 'Non-Hosp' THEN group = 1;  
      ELSE group = 0; *FORCES 'NON-HOSP' TO BE AT THE END ONCE SORTED;   
 
   PROC SORT DATA = report1;  
      BY group DESCENDING birth;  
   RUN; 
 
To further refine your summarized results, round the generated proportions and present them in a percentage form 
(i.e., 9.50% vs. 0.0950131234) in a second DATA step. The ARRAY statement  above utilizes the ROUND function 
to round the computed proportions to 4 digits after the decimal. The multiplication by 100 will display the data in a 
more readable format. Since larger medical centers are of more interest to users of such report (they submit larger 
numbers of birth certificates), you want to maintain them at the beginning of the report. But, at the same time, you 
may want to keep the “non-hospital” group at the bottom. To do that, create the temporary variable group (drop later) 
and then use PROC SORT to sort the data by group first  (variable will be sorted in ascending order by default, 
which keeps the string “non-hosp” after the numbers in a character variable), and then by the total number of births 
birth. Request a descending sort by placing the DESCENDING option before variable birth. This keeps hospitals with 
larger numbers of births at the top.  
 
STEP 3: COMPARE TO STANDARD 
In this DATA step, we create 3 new variables: (1) score85, that represents the number of variables in which a hospital 
rate of reporting non-classifiable, unknown, or blank data fields ranges from 85% to equal to standard; (2) score100, 
that represents the number of variables in which a hospital rate of reporting poor data exceeds NCHS standards; and 
finally, EBC, which indicates whether a hospital uses Electronic Birth Certificate System (EBC) or not.     
  
   DATA report2 (DROP = group i);  
      SET report1; 
      ARRAY scorex (8) gest1_mean -- wtgain1_mean;   
      ARRAY standard (8) _TEMPORARY_ (1 17.78 5.64 5.03 2.24 2.24 2.99 7.79);  
 
      score85  = 0;  
      score100 = 0;  
 
      DO i = 1 TO 8;  
         IF 0.85*standard (i) LE scorex (i) LE standard (i)  
         THEN score85 = score85 + 1;  
         ELSE IF scorex (i) GT standard (i)  
         THEN score100 = score100 + 1;  
      END; 
 
      IF bhosp_ IN ('HOSPITAL-01', 'HOSPITAL-02', 'HOSPITAL-03', 
                    'HOSPITAL-04', 'HOSPITAL-05', 'HOSPITAL-06', 
                    'HOSPITAL-0X');  
      THEN EBC = 'YES'; 
      ELSE EBC = 'NO';   
 
      IF bhosp_ EQ '' THEN DO;  
         bhosp_ = 'Total'; 
         EBC    = 'N/A';  
      END; 
 
      IF birth LE 5 THEN DELETE; 
   RUN; 
 
ARRAY scorex  catalogs the percentages of the 8 selected DATA step variables while standard  records the 
NCHS standard percentages. Note that we used the option _TEMPORARY_ to create the list of standard 
percentages as temporary data elements. You can create new variables that analyses the characteristics of hospitals 
such as rural vs. urban, secondary vs. tertiary, etc. Here, we are using the IF-THEN/ELSE argument  to classify 
hospitals as EBC vs. non-EBC hospitals. Delete hospitals reported less than 5 births during the period of evaluation.  

STEP 4: SET RANGES FOR COLOR-CODING USING PROC FORMAT    
The next FORMAT procedure assigns colors based on a decision rule2: orange if the percentage of a particular 
variable is 85% to equal to acceptable NCHS standards, and red if it exceeds NCHS standards.   
 
   PROC FORMAT;  
      value visits   2.54-2.99 = orange  3.02-high = red other = black;  
      value gest        0.85-1 = orange  1.01-high = red other = black;  
      value menda  15.11-17.78 = orange 17.96-high = red other = black;  
      value menmo    4.79-5.64 = orange  5.70-high = red other = black;  
      value menyr    4.28-5.03 = orange  5.08-high = red other = black;  
      value care     1.90-2.24 = orange  2.26-high = red other = black;  
      value nammo    1.90-2.24 = orange  2.26-high = red other = black;  
      value wtgain   6.62-7.79 = orange  7.87-high = red other = black; 
   RUN; 
 
STEP 5: APPLY THE FORMAT AND OTHER ODS STYLE DEFINITIONS WITHIN PROC REPORT  
 
   ODS LISTING CLOSE;  
   ODS DF FILP E = "&path\SUGI31.pdf" NOTOC STYLE = SASWEB;  
   PROC REPORT DATA = report2 NOWINDOWS                      
      STYLE (HEADER) = [FOREGROUND = YELLOW FONT_WEIGHT = BOLD FONT_SIZE = 8PT];  
      COLUMN bhosp_ birth EBC score100 score85 rec_err_sum rec_err_mean  
             gest1_sum gest1_mean menda1_sum menda1_mean menmo1_sum   menmo1_mean  
             menyr1_sum menyr1_mean care1_sum care1_mean nammo1_sum nammo1_mean  
             visits1_sum visits1_mean wtgain1_sum wtgain1_mean;  
 
Turn off the traditional LISTING destination and activate a PDF destination with a SASWEB STYLE , which by 
default, generates a white FOREGROUND in column headers (Figure 1). Include the NOTOC option in the ODS PDF 
destination statement to suppress the table of contents in your PDF file (Figure 2). With a STYLE statement , 
however, PROC REPORT gives you more control on column headers’ foreground color, font weight, and font size 
(Figure 3). Like a VAR statement in PROC PRINT, COLUMN statement  selects the arrangement of columns you 
want to present in your report.   
 
                            Figure 1                                                    Figure 2                                     Figure 3 
 
  

 
 
 
 
 
      DEFINE bhosp_/'Birth/Hospital';  
      DEFINE birth/'Birth/Count';  
      DEFINE score100/'Above/NCHS/Standard'  STYLE(HEADER) = [FOREGROUND = RED];  
      DEFINE score85/'85% to/Equal/Standard' STYLE(HEADER) = [FOREGROUND = ORANGE]; 
 
THE DEFINE statements above describe how to use and display a report item in PROC REPORT, such as labeling 
your columns. With the STYLE (ItemName) = option, you can even control style elements (in ODS) for report items 
such as column headers. Here, we are requesting the column header for the variable score100 be displayed in red, 
while column header for score85 be displayed in orange (Figure 4). This serves as a “legend” of how to read the 
color-coded values in the report.    
 
                            Figure 4                                                    Figure 5                                     Figure 6 

PROC STEP CONTINUED FROM PREVIOUS PAGE 
 
      DEFINE rec_err_sum/'Records/with at/Least/1 Error';   
      DEFINE rec_err_mean/'Percent';   
      DEFINE gest1_sum/'Clinical/Estimate/of/Gestation';  
      DEFINE gest1_mean/'NCHS/Standard/=1.00%'  
             STYLE (COLUMN) = [FOREGROUND = gest. FONT_WEIGHT = BOLD]  
             STYLE (HEADER) = [FONT_STYLE = ITALIC];  
      DEFINE menda1_sum/'Day/Last/Normal/Menses'; 
      DEFINE menda1_mean/'NCHS/Standard/=17.78%'  
         STYLE (COLUMN) = [ FOREGROUND = menda. FONT_WEIGHT = BOLD]  
         STYLE (HEADER) = [FONT_STYLE = ITALIC];  
      *Continue defining the rest of the variables the same way; 
 
      TITLE 'NATALITY NOT CLASSIFIABLE, UNKNOWN, AND BLANK DATA BY HOSPITAL: ARKANSAS 
             2003';  
   RUN;  
   ODS ODF CLOSE;  
   ODS LISTING; 
   QUIT; 
 
Recall that we did not use the AUTOLABEL option in PROC SUMMARY (step 1) because we wanted to take 
advantage of DEFINE statements in PROC REPORT. In this example, we are placing NCHS standards in the labels 
of the columns that contain the percentages. This works as a reference for comparison (Figure 5). Furthermore, this 
is where we apply the color-coding format  and , created in step 4, using the STYLE (COLUMN) =  
[FOREGROUND = FORMAT-Name] option (Figure 6).  Add a title to your report using the TITLE statement . 
Always place the RUN statement before ODS statements. Do not forget to turn off the ODS PDF destination and turn 
on the LISTING destination. 
 
THE FINAL PRODUCT 
The next 3 figures capture the generated report. The report is printed in a legal size paper with a landscape 
orientation. 

CONCLUSION 
In MCH and other reproductive health research and services, edited and complete birth data produces reliable and 
significant results. The Center for Health Statistics at the Arkansas Division of Health is highly rated from the National 
Center for Health Statistics (NCHS) regarding the accuracy, timeliness, and completeness of their data. To carry out 
quality MCH research for Arkansas citizens and to maintain these high ratings, the center evaluates the performance 
of birth hospitals concerning submitting invalid data in birth certificates. To accomplish that, they are putting SAS and 
its Output Delivery System (ODS) to use.          
 
The introduction of ODS motivated us to move from the multi-step SAS-Excel process of generating color-coded 
reports to an all-in-one process using PROC REPORT. The REPORT procedure combines features of the PRINT, 
MEANS, and TABULATE procedures with features of the DATA step in a single report-writing tool3. In this project, we 
are not using all of these features, nevertheless; the incorporation of color-coding formats into PROC REPORT is of 
exceptional utility for users of periodical reports in which certain levels of performance need to be highlighted.  
 
The default features provided by ODS in the different versions of SAS are convenient and can fit any report. To 
create customized reports, however, you may need to alter or modify defaults using the flexibility of ODS style 
definitions. PROC REPORT is an excellent venue to use these style definitions.       

